11-28
Глава 11. Промотор Большего Изображения

Изучение CA-Visual Objects 2.5


Глава 11

Просмотр  Большого Изображения
Рассмотрев предыдущий проект Video Tracking System (Глава 7), Вы помните, что мы создавали окна, помещали некоторые из доступных элементах управления на эти окна, и отображали данные в этих элементах управления. Для начала это было не плохо, но все, что мы могли - это видеть одновременно только одну запись. Есть большое изображение, которое мы не видим, оно станет нам доступным, когда мы научимся, использовать DataBrowser в наших прикладных программах. Но сначала, мы должны рассмотреть проекты и их копии.
Проекты

Цель Проектов состоит в том, чтобы отделить код в области диска и позволить больше чем одному разработчику работать с прикладной программой. 
Мы создадим новый проект, для этого щелкните по CA-Visual Objects 2.5 на левой панели. Выберите New Project в меню File, чтобы отобразить диалоговое окно New Project (Рис. 11.1), наберите "EVO" в Project Name, и имя каталога C:\EvoProj  в Project Directory и нажмите кнопку ОК.

	Рис. 11.1 Создавая новый проект укажите Имя проекта и Директорию


После этого должен создаться новый проект в новом каталоге (Рис. 11.2).

Копия ( Backup )

Есть ряд способов делать копии в Visual Objects. Все они включены в опцию Export в меню File. Экспортируемый элемент зависит от элемента, выбранного в Репозитарии. Например, откройте любую часть кода в Редакторе Исходного кода и выберите Export в меню File. 
Отобразится окно Сохранить как и Тип файла установлен на Source (*.PRG) (Рис. 11.3) который указывает, что он сохранит только эту часть кода. Нажмите кнопку Отмена, чтобы закрыть это окно. Закройте Редактор Исходного кода, чтобы возвратиться в Репозитарий. Установите курсор на модуль, который хотите экспортировать, и снова выберите Export в меню File. 

	Рис. 11.2 Новый проект в списке Visual Objects 2.5


На сей раз, Тип файла установлен в Mod. Export Files (*.MEF) и Имя файла уже установлено, которое имеет то же самое название что и модуль, который был выбран (Рис. 11.4). Он экспортировало бы все элементы в этом модуле.

	Рис. 11.3 Сохранение Исходного кода как файл .PRG, используя опцию Export из меню File


Снова, нажмите кнопку отмены, чтобы закрыть это окно. Последний тип копии - полный экспорт прикладной программы. Выберите Ch07 в левой панели и выберите Export в меню File. Тип файла установлен в App. (*.AEF) и Имя файла установлено в Ch07.AEF (Рис. 11.5). Установите сохранить в каталоге EVO\AEFS, и нажмите кнопку Save (Сохранить), чтобы экспортировать целую прикладную программу. Скоро мы будем использовать файл Ch07.AEF, но Вы могли бы рассмотреть экспорт всех других прикладных программ, которые Вы имеете в своем распоряжении. Для этого, Вы могли использовать опцию Export All в меню File.

	Рис. 11.4 Экспортируем модуль C:Start в Модульно Экспортный Файл (.MEF)

	Рис. 11.5 Экспортируем приложение Ch07 в Прикладной Экспортный Файл (.AEF)


Восстановление (Restore)

Нажмите на Проект EVO в левой панели репозитария, чтобы сменить Проект и выполните Import выбрав  в меню File. Перейдите в каталог EVO\AEFS, чтобы увидеть список доступных .AEF файлов (Рис. 11.6).

	Рис. 11.6 Импортирование приложения Ch07 в EVO


Выберите Ch07.AEF и нажмите кнопку Открыть, чтобы импортировать прикладную программу. Прикладная программа Ch07 появляется в левой панели репозитария, а все его объекты появляются в правой панели с Красными Крестами через их значки, указывающие, что они требуют компиляции (Рис. 11.7).

	Рис. 11.7 Приложения Ch07 показано красным крестом, указывающий, что оно требует компиляции.


Когда компиляция выполнится, щелкните правой клавишей на Ch07 в EVO выберите Rename и замените имя на "Ch11". Теперь выберите опцию Properties в меню Application и измените путь на %ExecutableDir%\Ch11.EXE; это позволит Вам пере импортировать Ch07.AEF в его первоначальную форму, если Вы желаете.

Мы можем проверить импорт, скомпилируйте и выполните прикладную программу. Когда Вы закончите проверять, закройте прикладную программу, чтобы возвратиться в Репозитарий.

Создание Видео браузера просмотра
Теперь, когда мы импортировали нашу Ch07 прикладную программу и переименовали её в Ch11, мы можем продолжить развитие прикладной программы. Сначала, создадим новый модуль по имени W:BrowseWin, откройте Редактор Окна (Window Editor), выберите Окно Данных (DataWindow) и введите имя "BrowseWin". На вкладке HyperLabel окна Свойств, в Caption пропишите ” Video Browser”. После этого перейдите на вкладку DataWindow, и измените свойство Width на 500 и Height на 272.
Мы предварительно видели браузер просмотра в окне (Рис. 3.23) но, что мы вероятно не замечали в то время,  так это, что требуется все окно. Не было никакого пространства   для других элементов управления подобно PushButtons или SingleLineEdits. Чтобы решить эту проблему, перейдите в окно Controls, выберите элемент управления Sub-DataWindow и перетащите контроль на пустую форму окна ближе к верхнему левому углу. На вкладке General окна Properties, установите name “SubBrowseWin”, Left на 7, Top на 11, Widtht на  486 и Height на 220. Наконец, нажмите на форме окна вне нового контроля Sub-DataWindow, чтобы удостовериться, что изменения вступили в силу (Рис. 11.8).

	Рис. 11.8 Элемент управления SubBrowseWin правильно измеряется и позиционируется на основе DataWindow


Сохраните окно и двойным щелчком нажмите на контроле SubBrowseWin относительно нашего Основного Окна, чтобы открыть второй Редактор Окна.

Первое, на что необходимо обратить внимание, что окно является уже с правильным размером, что соответствует области Основного Окна. Нажмите на кнопку AutoLayout, выберите Video2S Сервер в ComboBox и нажмите кнопку OK. Пока, мы выберем все поля  Сервера, нажмите кнопку OK.

Столбцы

Поля теперь отображены в виде Формы (Рис. 11.9) нажмите кнопку Browse:Form View, чтобы создать столбцы и окно просмотра (Рис. 11.10).

	Рис. 11.9 Поля сервера Video2S отображенные на SubBrowseWin в виде Бланка

	Рис. 11.10 Поля из Рис. 11.9 теперь отображены в виде Таблицы на SubBrowseWin


Вы можете выбрать любой столбец, нажимая на его область Заголовка и, как обычно, содержание окна Properties отражает в настоящее время выбранный элемент управления так, чтобы его свойства можно было изменять. Сначала, выберите столбец Vno, замените его свойство Caption на “No”, и протяните столбец направо, до 20mm .

Теперь, выберите столбец Title, и протяните его, до 60mm (Рис. 11.11).

Сохраните окно и закройте редактор, чтобы возвратиться к первоначальному Редактору Окна (то есть BrowseWin), закройте этот редактор также. Выберите модуль М:VideoShell,  и откройте Редактор Меню VideoShellMenu.

В Редакторе Меню, выберите второе меню, измените его свойство Caption  Open Video2, на “Browse Video” и свойство Event Name Property на “BrowseWin”. Сохраните изменения, закройте редактор, откомпилируйте, и выполните прикладную программу.

Выберите Browse Video в меню File (или нажмите кнопку Open) и здесь мы видим большее количество данных - или, по крайней мере, множество записей (Рис. 11.12).

	Рис. 11.11 

	Рис. 11.12 


Однако, мы видим что столбцы StarM, StarF и Category, показывают числа вместо соответствующего текста. Чтобы исправить это, мы должны закрыть прикладную программу и возвратиться в Репозитарий.

Идите в модуль W:BrowseWin и откройте SubBrowseWin. Мы нуждаемся в некотором разъяснении, чтобы разобрать два способа, как получить эту возможность просмотра.

Сначала, увеличьте ширину окна, перетаскивая правый край окна, и затем, когда Вы закончите, верните назад размер окна к первоначальным измерениям. Другой путь состоит в том, чтобы сократить более широкие столбцы к размеру, где Вы способны видеть все столбцы в окне; когда Вы закончили, Вы можете растягивать их, начиная с права. Оба из этих методов могут быть объединены, если требуется. Пока, обратите внимание на ширину окна и растяните окно вправо так, чтобы Вы могли видеть все столбцы. 

Посмотрите на окно Controls и Вы увидите, что есть только один контроль, который Вы можете выбирать – элемент управления Column. Нажмите на иконке элемента управления Column и перетащите на окно. Обратите внимание, что, независимо от того куда Вы перетаскиваете контроль, колонка всегда появляется в правом конце окна просмотра. Как только новый столбец находится на экране, перенесите его так, чтобы центр находился между  столбцами Title и StarM.

Когда новый столбец в фокусе, установите Caption в "Male Star” и измените Name на “MaleStar”. Теперь нажмите на столбец StarM и нажмите клавишу <Delete>, чтобы удалить его. Наступило время для испытания, чтобы посмотреть, где мы - поэтому сохраните окно, закройте редактор, откомпилируйте и выполните прикладную программу. 

Теперь, когда мы открываем окно Video Browser  (Рис. 11.13), мы видим, что столбец Male Star заполнен N/As, потому что нет никакого подключения к Серверу. Закройте прикладную программу, мы будем делать подключение.

	Рис. 11.13 


Идите в модуль C:Start и выберите New Entity -> Source Code Editor.

Введите следующий код- 

FUNCTION GetMaleStar(oVideoWin AS OBJECT) AS STRING

LOCAL cReturnString AS STRING

IF oVideoWin:MaleServer:Seek(oVideoWin:Server:StarM)

      cReturnString := AllTrim(oVideoWin:MaleServer:FName) ;

     + ’ ’ ;

     + AllTrim(oVideoWin:MaleServer:LName)

ELSE

     cReturnString := NULL_STRING

ENDIF

RETURN cReturnString

Перед тем как объяснить, что этот код делает, мы поместим остальную часть требуемого кода на место. Сохраните код, закройте редактора, возвратитесь к W:BrowseWin и откройте SubBrowseWin. Нажмите на вкладку DataWindow в окне Properties и откройте свойство Class Declaration. Здесь, мы получаем длинный список, PROTECT и INSTANCE переменных и в основании, мы имеем линию-

//{{%UC%}} USER CODE STARTS HERE (do NOT remove this line)

Эта линия очень важна, потому что все переменные выше управляются редактором. Если Вы делаете изменение в редакторе и затем сохраняете код, выше область этой линии, то код будет удален и перезаписан, так что в итоге любые изменения, которые Вы сделаете выше этой линия, будет потеряны. Это означает, что, чтобы добавить переменные к классу, они должны идти после этой линии.

Мы хотим добавить новую переменную после 

//{{%UC%}} USER CODE STARTS HERE (do NOT remove this line)

Напечатайте следующий код-

PROTECT oMaleServer AS StarM

Снова сохраните код, и закройте редактор. Теперь идите в основание вкладки DataWindow, и откройте свойство PreInit Actions. Измените код на-

METHOD PreInit(oWindow, iCtlID, oServer, uExtra) ;

CLASS SubBrowseWin

//Put your PreInit additions here

SELF:oMaleServer := StarM{ } // New Line

RETURN NIL
Снова, сохраните код, и закройте редактор. Теперь идите во вкладку Window Events (События окна), и откройте Редактор исходного кода для свойства Close. Измените код на-

METHOD Close(oEvent) CLASS SubBrowseWin

SELF:oMaleServer:Close() // New Line

SUPER:Close(oEvent)

//Put your changes here

RETURN NIL

Блок Столбца и Блок Владелеца
Сохраните код, закройте редактор и нажмите на колонке Male Star, чтобы колонка была в фокусе. В окне Properties, нажмите на вкладку General. Здесь мы имеем два свойства, которые связаны вместе, Block и Block Owner. В свойстве Block, введите –
{ | x | GetMaleStar (x)}

и в свойство Block Owner, введите –
SELF
Теперь сохраните окно и закройте редактор. Когда Вы закроете редактор, появится  диалоговое окно, предупреждая Вас, что сгенерированный код был изменен начиная с последнего сеанса. Это – произошло, потому что Вы добавили переменную PROTECT к декларации класса SubBrowseWin. Не волнуйтесь! Новая переменная была добавлена после 

строки ‘//{{%UC%}} USER CODE ...’ и поэтому она не будет перезаписана. Нажмите на YES.

Для последней части кода, в модуле C:Start выберите из меню New Entity -> Source Code Editor и введите следующий код –
ACCESS MaleServer CLASS SubBrowseWin

RETURN SELF:oMaleServer

Сохраните код, закройте редактор, и давайте поговорим о том, что мы сделали.

Поскольку столбец не связан с сервером, мы используем Block как путь заполнения столбца. Это написано как CodeBlock (Блок кода), о котором мы будем говорить более подробно позже (Глава 13).

Важная часть здесь - то, что 'x' между двумя вертикальными линиями является параметром, который используется в блоке кода. Этот параметр описан в функции GetMaleStar (x).

Очевиден вопрос : Что за значение будет проходить под <x>? Ответ, то что вложено в свойство Block Owner. Здесь, <SELF> передается как параметр Block Owner в блок кода и следовательно к GetMaleStar(). В этом случае, <SELF> - класс SubBrowseWin.

Блок кода работает для каждой записи в окне просмотра. Поэтому GetMaleStar() будет,  запрашивать каждую запись. Помните, что функция не принадлежит классу, так что нет никакой ссылки на SELF в ее коде. Единственный поступающий параметр - SubBrowseWin. Мы добавили PROTECT переменную к классу SubBrowseWin, чтобы создать сервер StarM, который мы открыли в его методе PreInit(). Который, в конечном счете закрывается в методе Close() - но Вы должны заметить, что он закрывается перед вызовом метода SUPER:CLOSE(), чтобы удостовериться, что сервер закрыт  прежде, чем главное окно уничтожено.

Вы наверное заметили, что мы полностью проигнорировали линию  // Put your changes here, потому что это - только подсказка. Декларация класса, метод Init() и Ресурс - единственные части кода, который будут записаны поверх, если изменения были сделаны в редакторе.

Поскольку функция не принадлежит классу, в ней нельзя видеть PROTECT переменную. Чтобы решить это, мы написали метод ACCESS, который возвращает переменную, делающую её видимой в GetMaleStar () функции.

Читая три введенные строки из функции GetMaleStar() справа налево, мы находим, что : Получаем значение StarM от сервера SubBrowseWin, и ищет это значение в MaleServer, принадлежащее SubBrowseWin.

Если значение <oVideoWin:Server:StarM> найдено, имя плюс пробел плюс фамилия сохраняется в <cReturnString>. Если не найдено, мы назначаем NULL_STRING в переменную. Это - константа, которая представляет пустое значение. Наконец, <cReturnString> возвращает в блок кода и следовательно к столбцу Male Star.

Приступим к нашему следующему испытанию. Откомпилируйте прикладную программу, и выполните её. Теперь мы можем видеть столбец Male Star гораздо более значащим. Что нам теперь нужно?. Мы должны сделать то же самое и для столбцов Female Star и Category для этого, нажмите на панели инструментов кнопку Close, чтобы закрыть, окно Video Browser,  нажмите на Close снова, чтобы закрыть оболочку.

Нажмите на модуль C:Start и откройте в редакторе функцию GetMaleStar(). Скопируйте весь код в буфер обмена, идите в основание кода, добавьте пустую линию и затем вставьте из буфера обмена. Теперь измените код  (изменения подчеркнуты) –
FUNCTION GetFemaleStar(oVideoWin AS OBJECT) AS STRING

LOCAL cReturnString AS STRING

IF oVideoWin:FemaleServer:Seek(oVideoWin:Server:StarF)

cReturnString := AllTrim(oVideoWin:FemaleServer:FName) ;

+ ’ ’ ;

+ AllTrim(oVideoWin:FemaleServer:LName)

ELSE

cReturnString := NULL_STRING

ENDIF

RETURN cReturnString

Теперь, добавьте пустую линию, и напечатайте следующий код –
FUNCTION GetCategory(oVideoWin AS OBJECT) AS STRING

LOCAL cReturnString AS STRING

LOCAL cCat AS STRING

cCat := oVideoWin:Server:Category

DO CASE

CASE cCat == ‘1’

cReturnString := ‘Action’

CASE cCat == ‘2’

cReturnString := ‘Drama’

CASE cCat == ‘3’

cReturnString := ‘Music’

CASE cCat == ‘4’

cReturnString := ‘Comedy’

OTHERWISE

cReturnString := NULL_STRING

ENDCASE

RETURN cReturnString

Сохраните изменения, и закройте редактор. Выберите модуль W:BrowseWin, откройте SubBrowseWin в Редакторе Окна и выборе вкладку DataWindow в окне Properties. Перейдите до свойства Class Declaration и откройте Редактор исходного кода, нажмите на кнопку с многоточиями в ячейке Value. Внизу кода есть линия, которую мы до этого добавили –
//{{%UC%}} USER CODE STARTS HERE (do NOT remove this line)

PROTECT oMaleServer AS StarM
Под этим, добавьте следующий код –
PROTECT oFemaleServer AS StarF
Сохраните код, и закройте редактор. Перейдите на вкладку Window Events (Свойства Окна)  в окне Properties, выберите свойство Close Event и откройте снова Редактор исходного кода,  нажав на кнопку с многоточиями. Измените код –
METHOD Close(oEvent) CLASS SubBrowseWin

SELF:oMaleServer:Close()

SELF:oFemaleServer:Close() // New Line

SUPER:Close(oEvent)

//Put your changes here

RETURN NIL
Сохраните код и закройте редактор. Прежде, чем мы закончим с Редактором Окна, мы должны разобраться в столбцах. Нажмите на заголовок столбца StarF и измените свойство Name на "FemaleStar". Это действие нарушит связь с данным, потому что Name больше не соответствует имени поля сервера. Измените свойство Caption на “Female Star”. Перейдите на вкладку General и установите свойство Block в –
{ |x | GetFemaleStar (x)}
и свойство Block Owner –
SELF
Теперь, нажмите на заголовок столбца Category, чтобы заголовок был в фокусе,  и измените свойство Name на "CatCol". Установите свойство Block –
{ |x | GetCategory (x)}

и Block Owner –
SELF
Наконец, нажмите на столбце FemaleStar, переместите мышь в правый край Header до изменения указателя, и растяните столбец, до размера столбца MaleStar. Это может сместить столбец Category за края окна, но не беспокойтесь об это в данный момент. Сохраните окно, проигнорируйте предупреждение о сгенерированном коде, и закройте Редактор Окна, чтобы возвратиться в Репозитарий. Убедитесь что C:Start в фокусе, выберите Source Code Editor  в меню New Entity и введите следующий код –
ACCESS FemaleServer CLASS SubBrowseWin

RETURN SELF:oFemaleServer
Сохраните код и закройте редактор. Возвратитесь к модулю W:BrowseWin и нажмите двойным щелчком на методе SubBrowseWin:PreInit(). Он откроется в Редакторе исходного кода. Измените код на –
METHOD PreInit(oWindow, iCtlID, oServer, uExtra) ;

CLASS SubBrowseWin

//Put your PreInit additions here

SELF:oMaleServer := StarM{ }

SELF:oFemaleServer := StarF{ } // New Line

RETURN NIL
Сохраните код и закройте редактор. Мы сделали несколько ссылок на Female Server который, в этом модуле не существует, поэтому мы должны установить эту связь следующим образом. Создайте новый модуль, назовите "S:Starf " и откройте DBServer Editor для нового модуля. Нажмите кнопку File Find, измените каталог на C:\EVO\Video, и выбирите файл Starf.DBF. Теперь, нажмите на кнопку Index Find и выберите Starf.NTX, чтобы закончить сервер.

Сохраните сервер, закройте DB Server Editor, нажмите кнопку Build и, когда она закончится, нажмите кнопку RUN. 

Когда Вы нажмете кнопку Open на Панели инструментов и отобразится окно Video Browser, Вы заметите две вещи (Рис. 11.14).

	Рис. 11.14 


Во первых - то, что столбец FemaleStar не показывает названия. Во вторых - то, что, поперек основания окна просмотра появился Scroll Bar. Scroll Bar автоматически вставляется, как только окно становится слишком большим, чтобы отобразиться на всем экране.

Используйте Scroll Bar, чтобы сдвигаться вправо, и Вы увидите, что только первые два символа текста категории отображены в столбце Category. Подобная проблема и в отображении  столбца FemaleStar.

Чтобы найти причину, мы должны возвратиться в редактор, поэтому дважды нажмите кнопку Close на Панели инструментов. Идите в модуль W:BrowseWin, и откройте SubBrowseWin в Редакторе Окна. Что мы ищем,  - различие между столбцом MaleStar, который работает правильно и двумя другими, что мы только что изменили. Выберите столбец MaleStar и нажмите на вкладку General. Вы увидите, что свойство Field Spec  установлено в <Auto>. Выберите столбец FemaleStar и мы видим, что в том же свойстве установлено Video2S_Starf. Это - потому что, когда столбцы были созданы для нас, свойства были приняты от сервера файла данных, которым является Video2S. Если Вы посмотрите на столбец CatCol, то Вы найдете в свойстве Field Spec тоже самое. 
Исследуем это немного, прежде, чем мы начнем исправления. Закройте Редактор Окна, перейдите в модуль S:Video2S и откройте для Video2S_Starf Редактор FieldSpec (Рис. 11.15). (Есть два данных с Заголовком Video2S_STARF в правой области Репозитария. Вы увидите, что первый из них - Декларация Класса, а второй - FieldSpec. Дважды щелкните на значке FieldSpec, чтобы открыть редактор FieldSpec.)

	Рис. 11.15 


Они являются правилами, которые применяются к столбцу и здесь мы можем видеть, что свойство Length установлено в '5', которая показывает, почему столбец FemaleStar усекается.

	Рис. 11.16


Проблема состоит в том, что, если мы изменим его здесь, он не будет работать, если он используется на сервере, для которого он был создан. Есть два пути решения этой проблемы. Первый - ВЫ должны создать новый FieldSpec и прикрепить его к столбцу; второй переписать FieldSpec от столбца и указать значения, которые управляют этим столбцом. Мы будем использовать второй вариант, для этого закройте Редактор FieldSpec, (не волнуйтесь, мы охватим это подробно в Главе 16), идите в модуль W:BrowseWin и откройте окно SubBrowseWin в Редакторе Окна. 
Выберите столбец FemaleStar, перейдите на вкладку General и установите свойство Field Spec в <Auto>, опция наверху списка. Теперь cделайте тот же самый для столбца CatCol.
Сохраните окно, закройте редактор, скомпилируйте, и выполните прикладную программу. 
Теперь, когда Вы откроете Video Browser (Рис. 11.16), что столбец Female Star правильные имена и если вы используете ScroolBar, чтобы переместится вправо, то столбец Category должен так же показывать правильные данные.
Пока мы - здесь, попробуйте дважды щелкнуть мышью в одной из ячеек в столбце Category.

	Рис. 11.17 


Вы увидите (Рис. 11.17) что текст перешел  в Режим Редактирования.  
Причина в том что, когда мы вносили изменения в столбец Category - нарушили связь с файлом данных, поэтому любые изменения, которые мы делаем, не будет сохранены. Но колонки, с которыми установлена связь, могут редактироваться, что не очень хорошо. 

Есть несколько путей решения этой проблемы, но самый легкий, мы должны сделать окно просмотра READ ONLY, чтобы мы не могли изменять данные в режиме редактирования. Мы это сейчас и сделаем. Нажмите дважды кнопку Close на Панели инструментов, чтобы закрыть прикладную программу и закончить испытание.

Окно просмотра Только для чтения

Найдите метод  SubBrowseWin:Init() в модуле W:BrowseWin, откройте  Редактор Исходного кода, и переместитесь до линии 63, кода –
SELF:Browser: = DataBrowser {SELF}

Это - инициализация Окна просмотра и, отсюда вниз, инициализация столбцов

Запомните, что мы хотим сделать так, чтобы нельзя было свободно делать любые изменения. Что мы сделаем, мы должны найти способ  создания окна только для просмотра. Закройте редактор, и повторно откройте его через меню New Entity. Теперь введите следующий код –
CLASS VideoROBrowser INHERIT DataBrowser

METHOD Init(oOwner) CLASS VideoROBrowser

SUPER:Init(oOwner)

SELF:SetStandardStyle(GBSREADONLY)

RETURN SELF
Сохраните код, закройте редактор, и нажмите на кнопку Build. Когда компиляция закончена, откройте SubBrowseWin в Редакторе Окна и щелкните по вкладке  DataWindow в Окне Свойств (Properties Window). 
Свойство Browser Inherits From, установлено в <Аuto>, но, когда Вы выберите его, то вниз-спадающем списке параметров будет доступен наш новый класс VideoROBrowser, чтобы быть выбранным. Нажмите на эту Опцию, сохраните Окно, закройте редактор, скомпилируйте, и выполните прикладную программу. Теперь, когда Вы откроете окно Video Browser и пробуете редактировать ячейку, дважды щелкнув по полю, ничего не происходит. Закройте испытание, и давайте смотреть то, что мы сделали, как заставили это работать. 
Снова, найдите метод SubBrowseWin:Init() в модуле W:BrowseWin, откройте его в Редакторе исходного кода и спуститесь к линии 63, где Вы увидите эту линию кода –
SELF:Browser: = VideoROBrowser {SELF}

Она - теперь инициализируется как Browser. Код, который мы ввели, указывает, что VideoROBrowser наследуется от DataBrowser так, чтобы, когда наш метод Init()  выполнился, запрос SUPER фактически вызывает метод Init (), который был бы выполнен, если нашего класса не было там. После того, как SUPER:Init () закончился, мы включаем нашу новую установку, которая делает окно просмотра только для чтения. 
Это - чистая демонстрация того, как изменяется поведение, и что Вы видите на экране, используя наследование, чтобы создать новый класс от существующего без того, чтобы изменять оригинал. Закройте редактор, чтобы возвратиться в Редактор Окна.

Давайте совершим быстрый обзор прежде, чем мы двинемся дальше. Первоначально мы создали окно по имени Video2, которое показывало нам одну запись и давало нам возможность двигаться по записям. Теперь мы создали окно, которое показывает несколько записей и легко перемещаемся между ними, без какого либо шанса изменить данные. 

Повторное посещение Video2 

В настоящее время, мы не используем окно Video2, но, в следующем примере, мы будем работать с предположением, что, где-нибудь в прикладной программе, окно Video2  потребуется почти таким каким оно есть.  Причина по которой мы говорим 'почти' – в том, что способ, которым мы заполняем элемент управления ComboBox MaleStar является неудовлетворительным. Опция Use Server - не соответствует для этой ситуации, потому что она только показывает поле LName принимая во внимание, что элемент управления ComboBox FemaleStar показывает оба имени, имя и фамилию. Это необходимо изменить, чтобы сделать окно Video2 пригодным к использованию, так что мы в следующем задании, будем исправлять это проблему.

Идите в модуль W:Video2 и откройте Video2 в Редакторе Окна. Прежде всего, измените его свойство Caption на " Edit Video ". Выберите вкладку DataWindow и измените свойство Menu y на <Auto> в окне Properties. Затем, выберите элемент управления ComboBox Male Star и нажмите на многоточии в  свойстве Fill Using. Нажмите на радио-кнопке Use Method, установите тип "MaleStarFill" и нажмите кнопку OK. Сохраните окно, и закройте редактор. Идет в модуль C:Video2 и откройте метод FemaleStarFill()  в Редакторе исходного кода. Копируйте весь код в буфер обмена, закройте редактор и откройте его снова в опции New Entity. Вставьте скопированный код в редактор и измените декларации метода на –
METHOD MaleStarFill() CLASS Video2
и измените строки, которые открывают файлы данных и индексов –
oServer := DBServer{"C:\EVOVideo\StarM.dbf"}

oServer : SetIndex("C:\EVOVideo\StarM.ntx")

Сохраните код и закройте редактор. 
Нам нужно кое-что изменить в ShellMenu так, чтобы мы были способны вызвать окно Video2. Идите в модуль М: VideoShell и откройте VideoShellMenu. Выберите вторую строку и нажмите <Enter>, чтобы добавить пустую линию между опциями Browse Video и Close Window. Наберите " Video Form " на новой линии, потом идите в свойство Event Name  и наберите "Video2Load". Перейдите до свойство Button Bmp, откройте окно  Toolbar Buttons, выберите битовое изображение New (в конце списка) и нажмите кнопку OK. Измените свойство Button Pos  на "2", сохраните меню и закройте редактор.  

Последний шаг который  мы должны сделать связать  Меню с Окном, поэтому, создайте новый модуль, назовите "C:ShellWindow", откройте Редактор исходного кода и наберите следующий код –
METHOD Video2Load() CLASS ShellWindow

LOCAL oWin AS Video2

oWin := Video2{SELF}

oWin:Menu := Video2M{ }

oWin:Show()

RETURN NIL
Здесь мы добавляем метод к функциональным возможностям класса ShellWindow. Единственное, что необходимо помнить при добавлении новых функциональных возможностей к классу, то, что Вы можете отменить метод, который уже существует. Этот метод подобен тому что мы написали прежде, но на сей раз мы имеем новый код. Вместо создания окна, которое уже знает о меню, которое предполагается использовать, мы создаем окно и затем сообщаем ему меню которое оно должно использовать. Есть несколько причин для этого выполнения. 
Прежде всего, вообразите ситуацию, где Вы имеете три уровня защиты. Вы могли использовать то же самое окно с одним из трех различных меню в зависимости от уровня защиты пользователя. Другая причина для этого выполнения; этот способ состоит в том, что мы собираемся создавать связь с окном Video Browser и формой Edit Video для редактирования целей и не хотим, чтобы пользователь мог использовать меню, чтобы перемещаться по записями.

Browser:CellDoubleClick ()

Сохраните код, закройте редактор, идите в модуль W: BrowseWin, откройте Редактор исходного кода и напишите следующий код –
METHOD CellDoubleClick() CLASS VideoROBrowser

LOCAL oWin AS Video2

oWin := Video2{SELF:Owner:Owner:Owner, ,SELF:Owner:Server}

oWin:Show()

RETURN NIL
Сохраните код, и давайте посмотрим, что мы здесь имеем. Этот код добавляет к функциональным возможностям нашего окна Video Browser, обеспечивая метод, который нужно использовать, когда мышь дважды нажата в ячейке. Это - метод CallBack, что означает, что мы не должны вызывать его непосредственно, потому что окно просмотра вызовет его для нас в нужное время автоматически. Метод просто создает окно Video2 и передает окну три параметра. 
Первый параметр - владелец нового окна, и мы должны быть внимательными, что мы выбираем, потому что поведение окна будет зависеть от владельца. В этой ситуации <SELF> - окно просмотра, владелец окна просмотра - SubBrowseWin, владелец SubBrowseWin - BrowseWin, и владелец BrowseWin - ShellWindow. Все это означает, что, формированием цепочки назад через собственность объектов, мы в конечном счете приходим к ShellWindow как к владельцу нового окна. 
Третий параметр - Сервер, чтобы использовать при показе окна. Поскольку мы уже используем сервер, чтобы отобразить окно просмотра, это бессмысленное потребление памяти, чтобы позволить новому окну открывать новый нормальный сервер. Вместо этого, мы можем передавать сервер, который уже создан, что экономит память, получаем добавленное преимущество в том, что  редактирование - уже на правильной записи. В этой ситуации, <SELF> - окно просмотра, владелец окна просмотра - SubBrowseWin, и свойство  SubBrowseWin - сервер, который мы хотим использовать. Когда Вы поняли эту концепцию, закройте редактор. 
Прежде, чем мы выполним наше следующее испытание, мы должны сделать сервер Video2S  общедоступным так, чтобы мы могли использовать его больше, чем для одного пользователя. Снова, самый легкий способ сделать это состоит в том, чтобы установить свойство Exclusive, так что идите в модуль C: Start и откройте метод App:Start(). Теперь, измените начало метода –
METHOD Start() CLASS App

LOCAL oShell AS ShellWindow

SetExclusive(FALSE) // New Line

aCategory := { }

aCategory := FillCategory()

Сохраните код, закройте редактор, откомпилируйте, и выполните прикладную программу.
Нажмите левую кнопку на Панели инструментов  и мы увидим наше окно Video Browser как прежде. Но теперь, когда мы дважды щелкаем на записи появляется окно с данными записи, которую мы готовы редактировать. Перетащите окно Edit Video вправо так, чтобы Вы могли видеть также окно Video Browser и нажмите на одну из кнопок перемещения на окне Edit Video (Рис. 11.18).

	Рис. 11.18 


Вы увидеть, что записи не только изменяется в окне Edit Video, но также и в окне Video Browser. Это -  потому что они, совместно используют тот же самый сервер. Это работает, потому что, когда Указатель Записи на сервере перемещается, сервер посылает сообщения другим объектам, сообщая им, что произошли изменения, чтобы они могли принять меры соответственно. Мы увидим гораздо больше это чуть позже. 
Только, чтобы доказать эту систему передачи сообщений – есть два способа,  нажмите на различные записи в окне Video Browser. Это действие переместит окно Edit Video в фоновый режим, но, когда Вы вернетесь в таблицу, Вы увидите, что запись, кажется не изменилась. Однако, если Вы нажимаете одну из кнопок перемещения, чтобы перейти к предыдущей или следующей записи, Вы увидите, что они синхронизировались снова. Причина,  почему они,  выходили из синхронизации заключается в том что элемент управления редактирования отсоединялось от сервера, так что мы были должны обеспечить путь обновление экрана всякий раз, когда Указатель Записи на сервере перемещается. В нашей текущей ситуации, мы просто хотим показать способность перемещаться, в то время когда мы находимся в окне Edit Video,  редактируется запись, когда Вы закончили проверять, закройте прикладную программу, чтобы возвратиться  в Репозитарий. 

Скрытие элемента управления 

Чтобы выполнить следующий раздел, идите в модуль C:Video2, откройте Редактор исходного кода  через всплывающее меню New Entity и напишите следующий код –
METHOD SetForEdit() CLASS Video2

SELF:oCCGoTop:Hide()

SELF:oCCSkipPrevious:Hide()

SELF:oCCSkipNext:Hide()

SELF:oCCGoBottom:Hide()

SELF:oCCpbClose:Hide()

RETURN NIL

Этот метод почти, сам объясняет что он делает. Он вызывает методом Hide()  для каждого из элемента управления PushButton, которое принадлежит этому окну и возвращает NIL. Нам нужно указать месте вызова этого метода, сохраните код и закройте редактор, измените модуль W:BrowseWin, откройте метод CellDoubleClick() класса VideoROBrowser и изменение  код –
METHOD CellDoubleClick() CLASS VideoROBrowser

LOCAL oWin AS Video2

oWin := Video2{SELF:Owner:Owner:Owner, ,SELF:Owner:Server}

oWin:SetForEdit() // New Line

oWin:Show()

RETURN NIL
Как Вы вероятно догадываетесь, он выполняет скрытие PushButtons прежде, чем окно появится. Сохраните код, закройте редактор, откомпилируйте и выполните прикладную программу. 
Когда Вы откроете окно Video Browser,  нажмите двойным щелчком на записи, откроется окно Edit Video и Вы увидите что нет никакого PushButtons. Это, однако, не сделало его более полезным. Любые изменения сделанные в отображаемых данных потеряны, потому что, у нас нет никакого способа, модифицировать файл данных. Когда Вы закончите проверять, закройте прикладную программу. Теперь перейдем к решению это проблемы. Сначала, перейдите  в модуль W:Video2 и откройте первый объект, который должен быть декларацией класса. Внизу этого кода должна быть строка –
//{{%UC%}} USER CODE STARTS HERE (do NOT remove this line)
Ниже этой линии добавьте код –
PROTECT oCCpbSave AS PUSHBUTTON

PROTECT oCCpbCancel AS PUSHBUTTON

PROTECT oCCpbNew AS PUSHBUTTON

PROTECT lNewRecord AS LOGIC

Сохраните этот код, и закройте редактор. Перейдите в модуль C: Video2, откройте метод SetForEdit() и измените код -

METHOD SetForEdit() CLASS Video2

SELF:oCCGoTop:Hide()

SELF:oCCSkipPrevious:Hide()

SELF:oCCSkipNext:Hide()

SELF:oCCGoBottom:Hide()

SELF:oCCpbClose:Hide()

// Start of New Section ...

SELF:oCCpbCancel := PushButton{SELF, 601, Point{20, 20}, ;

Dimension{50, 20}}

SELF:oCCpbCancel:HyperLabel := HyperLabel{#pbCancel, ;

"Cancel", NULL_STRING, NULL_STRING}

SELF:oCCpbCancel:Show()

SELF:oCCpbNew := PushButton{SELF, 602, Point{320, 20}, ;

Dimension{50, 20}}

SELF:oCCpbNew:HyperLabel := HyperLabel{#pbNew, ;

"New", NULL_STRING, NULL_STRING}

SELF:oCCpbNew:Show()

SELF:oCCpbSave := PushButton{SELF, 603, Point{380, 20}, ;

Dimension{50, 20}}

SELF:oCCpbSave:HyperLabel := HyperLabel{#pbSave, ;

"Save", NULL_STRING, NULL_STRING}

SELF:oCCpbSave:Show()

// ... End of New Section

RETURN NIL

Сохраните код, и давайте посмотрим, что мы написали. Прежде всего, мы добавили три переменные к классу Video2, две переменные объявлены как PushButtons и одна переменная как LOGIC. Затем, мы, добавили код, чтобы создать три PushButtons, установили их HyperLabels и затем Show(). 
Рассмотрим параметры, которые получает класс PushButton, первый - владелец, которым является окно, второй - уникальный номер, который, используется внутри класса (выбор 601, и т.д. вполне произволен), треть - объект точки, который представляет левый угол основания элемента управления и четвертый - объект размер контроля. 
При создании класса HyperLabel, мы передаем четыре параметра: первый - символьное название элемента управления, второй - заголовок, третий - описание элемента управления и четвертый - строка контекстной справки, которую мы рассмотрим позже. 
Время для испытания близко, откомпилируйте и выполните прикладную программу. Теперь, когда Вы открываете окно Video Browser и два раза щелкаете на записи, откроется окно Edit Video и появятся созданные PushButtons, но когда - щелкаете на PushButtons ни ничего не происходит. Это потому что, нет никаких методов выполняющих эту работу. Закройте окно, чтобы закончить испытание. Щелкните правой кнопкой на модуле C:Video2, выберите Source Code Editor в всплывающем меню New Entity  и ведите следующий код –
METHOD pbNew() CLASS Video2

SELF:oDCsleVideoNo:TextValue := NULL_STRING

SELF:oDCslePosition:TextValue := NULL_STRING

SELF:oDCsleTitle:TextValue := NULL_STRING

SELF:oDCslePlayTime:TextValue := NULL_STRING

SELF:oDCcbMaleStar:CurrentItemNo := 0

SELF:oDCcbFemaleStar:CurrentItemNo := 0

SELF:oDCcbCategory:CurrentItemNo := 0

SELF:lNewRecord := TRUE

SELF:oDCsleVideoNo:SetFocus()

RETURN NIL

METHOD pbCancel() CLASS Video2

SELF:EndWindow()

RETURN NIL
METHOD pbSave() CLASS Video2

IF SELF:lNewRecord

SELF:Server:Append()

ENDIF

SELF:Server:VNo := SELF:oDCsleVideoNo:TextValue

SELF:Server:Pos := SELF:oDCslePosition:TextValue

SELF:Server:Title := SELF:oDCsleTitle:TextValue

SELF:Server:PlayTime := SELF:oDCslePlayTime:TextValue

SELF:Server:StarM := SELF:oDCcbMaleStar:Value

SELF:Server:StarF := SELF:oDCcbFemaleStar:Value

SELF:Server:Category := SELF:oDCcbCategory:Value

SELF:Server:Commit()

SELF:EndWindow()

RETURN NIL
Сохраните код, и давайте рассмотрим то, что он делает. 
Метод pbNew(), устанавливает для каждого элемента управления пустое значение. Затем мы можем увидеть использование <lNewRecord>, четвертая переменная, которую мы добавили к классу после включения переменных, для нажатия клавиш. <lNewRecord> установлен TRUE так, чтобы позже мы знали, что делать со значениями, введенными в элементы управления. Наконец, когда мы устанавливаем значение элемента управления, фокус автоматически перемещается в тот элемент управления который, в этом случае, должен поместить нас в combobox Category. Было бы очень неестественно запустить редактирование от последнего поля на форме, так что мы задерживаем фокус в первом элементе управления относительно экрана перед возвращением элемента управления пользователю. 
Метод  pbCancel() просто закрывает окно, отказываясь от любых изменений, потому что они не были записаны в файл. 
Метод  pbSave() проверяет  что мы делаем, или это должна быть новая запись или изменение текущей записи. Если была нажата кнопка New , <lNewRecord> будет TRUE, и мы вызываем метод Append()  добавляя пустую запись в файл данных. Она автоматически становится текущей записью и мы можем поэтому заменить пустые поля введенными данными. Как только данные занеслись, мы  вызываем метод Commit(), который вынуждает данные записываться на диск. Наконец, мы закрываем окно Edit Video так, чтобы мы могли видеть снова окно Video Browser. Время провести испытания, откомпилируйте, и выполните прикладную программу. 
Когда откроется окно Video Browser, обратите внимание на запись и затем дважды щелкните на ней, чтобы открыть окно Edit Video. Измените значения записи, нажмите кнопку Save. Когда окно Edit Video закроется, Вы должны увидеть, что запись была модифицирована и это отобразилось в окне Video Browser. Это из-за системы передачи сообщений, которая сообщила окну просмотра, что запись изменилась и окно должно обновится. 

	Рис. 11.19 


Затем, дважды щелкните на любой записи и нажмите кнопку New в окне Edit Video, чтобы очистить поля, для нового ввода (Рис. 11.19). Введите следующие значения –
Video No: 12

Position  : 1

Title        : New Tape
Нажмите кнопку Save PushButton.

	Рис. 11.20 


Форматирование Данных

Мы теперь видим нашу новую запись (Рис. 11.20), но Вы возможно заметили  две вещи. Сначала, запись находится в верхней области обзора, вместо позиции согласно Video No. Во-вторых, смотря на столбцы No и Position, мы видим, что значения смещено относительно других записей. Данные, который были введены ранее выровнены в право, как если бы они были числовыми столбцами. Закройте прикладную программу, и откройте снова метод pbSave(). Теперь измените первые две строки кода на –
SELF:Server:VNo := ;

STR2(VAL(SELF:oDCsleVideoNo:TextValue), 4)

SELF:Server:Pos := ;

STR2(VAL(SELF:oDCslePosition:TextValue), 4)

Берёте значение TextValue элемента управления, преобразовываем значение в число, используя функцию Val() и затем преобразовываем значение назад в строку с длиной до четырех символов, используя функцию STR2(). Сохраните код, закройте редактор, откомпилируйте и выполните прикладную программу. 
Откройте окно Video Browser, щелкните двойным щелчком на записи New Tape, которую  мы добавили в нашем прошлом испытании.  Вы найдете запись в  правильной позиции в Tape Number, этому есть короткое объяснение. Данные отобразились, поскольку мы этого  ожидали и мы не должны делать любых перемещений, поскольку они будут сделаны для нас автоматически, когда мы нажимаем кнопку Save. 
Когда окно Edit Video закрывается, Вы должны увидеть модифицированную  запись и значения входят в строку с остальной частью столбца. Это исправило проблему выравнивания, но не исправило неправильное позиционирование. Закройте прикладную программу и повторно откройте метод pbSave() для того, чтобы мы могли видеть то, что получилось. 
Мы должны понять то, какие сообщения посылаются от сервера к окну просмотра. Когда новая запись добавлена в конец, окно просмотра получает сообщение,  что требуется обновить - но в этой точке, индексный ключ записи не обновился, и поэтому запись устанавливается сверху окна просмотра в индексном порядке - хотя фактически запись находится внизу файла в естественном порядке. Когда значения помещены в запись, окно просмотра получает другое сообщение, которое говорить, что запись, изменилась и что необходимо обновить запись. Это оставляет нас с проблемой, которая сейчас показывает нашу запись в неправильном месте и мы решим это следующим образом, добавим строку кода в конец  метода pbSave() 

SELF:Server:Commit()

SELF:Server:Notify(NOTIFYFILECHANGE)//Новая строка
SELF:EndWindow()

В новой строке, мы вызываем метод Notify() для нашего сервера и передаем значение NOTIFY FILE CHANGE (обратите внимание, что в коде, она написана без пробелов между словами). Метод Notify(), фактически Callback Method и вызывается автоматически Классом Сервера, чтобы передать его сообщения. Это - один из немногих случаев, когда мы вызовем метод Notify() в нашем коде, но это – нужно сделать, так как все функциональные возможности для передачи сообщений встроены в сервер, и мы просто требуем, чтобы дополнительное изменение было послано. Изменение файла заставит обновить полностью окно просмотра и поэтому установит запись правильно. Чтобы проверять это, сохраните код, закройте редактор, откомпилируйте и выполните прикладную программу. Откройте окно Video Browser  двойным щелчком на любой записи. Нажмите кнопку New в окне Edit Video,  чтобы очистить поля и введите следующие значения –
Video No : 13

Position : 1

Title    : New Tape2 

и нажмите кнопку Save. 
Теперь мы видим нашу новую запись в правильном месте (Рис. 11.21), но, если Вы обратили внимание,  что запись сначала появляется в верхней части  окна просмотра, а затем перемещается в правильную позицию как отдельная операция. Это - эффект издания сообщения Notify. Когда Вы закончили добавлять и редактировать записи, закройте прикладную программу, чтобы перейти нам к концу нашего последнего испытания, чтобы мы могли посмотреть то, чего мы достигли.

	Рис. 11.21 


Обзор

Мы охватили весьма небольшой материал в этой главе. Мы рассмотрели, как создавать новый Projects и выполнять Backups наших прикладных программ. 
Мы  продолжили рассмотрение Browsers и Columns. С использованием свойств Block и Block Owner, когда мы создавали столбцы, которые непосредственно не были связаны с полем данных. 
Мы создали Browser Read Only и показали чистую демонстрацию наследования. 
Затем Мы рассмотрели использование Callback Methods, в этом случае, использовали метод CellDoubleClick() при открытии другого окна. 
При выполнении этого мы показывали, что Chaining Properties of Classes, Вы можете видеть, кто владеет окном и какой сервер используется. 
При выполнении этого, мы видели, что вид окна может быть изменен во время выполнения, скрывая некоторое средство управления и создавая новые так, чтобы способности окна были полностью различны. 
Наконец, мы рассмотрели форматирование данных при сохранении записи, чтобы показать непротиворечивость в файле данных и мы получили небольшое понимание в Системе Передачи сообщений, которую мы охватим более подробно позже.

И последнее 

Прежде, чем мы закончим эту главу, мы должны рассмотреть ситуацию с окном Video2. Мы теперь имеем одно окно с двумя целями. Первое - главное, добавляет  данные с немногими ограничениями. Второе гораздо более ограничено с некоторыми удаленными функциональными назначениями и другими добавленными частями. Чтобы получить этот эффект, мы как программисты, не должны забывать вызывать метод SetForEdit(). Но что было бы если нам нужно было поместить дополнительную кнопку на экране? 
Возможно было бы лучше наследовать класс Video2 и поместить запрос к методу SetForEdit()  в Init () того класса. Тогда, методы pbCancel(), pbNew() и pbSave(), также как SetForEdit(), могли быть перемещены в новый класс и были бы доступны любому экземпляру этого класса. 
Лучше всего было бы создать одно окно, только с общими элементами управления. Мы могли тогда наследовать это окна дважды, добавляя требуемые элементы управления в каждый класс. Это убирает потребность в сокрытии некоторых элементов управления.
Как испытание того, что Вы узнали, Вы можете пробовать сделать это. Если так, не забудьте снять копию .AEF. Вы можете всегда удалить прикладную программу и импортировать .AEF, чтобы возвратиться снова, где Вы находились. Удачи!
